Formulation of a Simple Market Clearing

- **Demand:** $P \geq Pd = 6 - 0.3*Qd$
- **Supply:** $P \leq Ps = 1 + 0.2*Qs$
- **Equilibrium:** $Qs \geq Qd$ and $P, Qs, Qd \geq 0$

2 commodities: corn and wheat

Corn Demand: $P_c \geq Pd_c = 6 - 0.3*Qd_c - 0.1*Qd_w$

Wheat Demand: $P_w \geq Pd_w = 8 - 0.07*Qd_c - 0.4*Qd_w$

Corn Supply: $P_c \leq Ps_c = 1 + 0.5*Qs_c + 0.1*Qs_w$

Wheat Supply: $P_w \leq Ps_w = 2 + 0.1*Qs_c + 0.3*Qs_w$

Corn Equilibrium: $Qs_c \geq Qd_c$

Wheat Equilibrium: $Qs_w \geq Qd_w$

$P_c, P_w, Qd_c, Qd_w, Qs_c$, and $Qs_w \geq 0$

Formulation of a Simple Market Clearing

Set Definition & Data Entry

```
SET Commodities  commodities used in the model          /Corn,Wheat/     ;
SET Curvetype    supply and demand intercept and slope  /Supply,Demand/ ;

TABLE intercepts(Curvetype,Commodities)  supply and demand intercept terms
                  Corn   Wheat
        demand     6       8
        supply     1       2;

TABLE Slopes(Curvetype,Commodities,Commodities)  supply and demand slope te
                  Corn  Wheat
    Demand.Corn   -.3    -.1
    Demand.Wheat  -.07   -.4
    Supply.Corn    .5     .1
    Supply.Wheat   .1     .3     ;
```

Set Definitions

In algebraic modeling, we commonly have subscripts.

In GAMS, the corresponding items are sets. A set definition has several potential parts.

```
SET    ItemName        optional explanatory text for item
     / element1        optional explanatory text for element ,
       element2        optional explanatory text for element / ;
```

SET or SETS	to start
ItemName	a unique name
optional explanatory text for item	
/	opening slash
Element names	
optional explanatory text for element	
, or line feed	to separate elements
/	closing slash
;	a closing ;

Set Definitions

In our example:

```
SET Commodities  commodities used in the model          /Corn,Wheat/    ;
SET Curvetype    supply and demand intercept and slope  /Supply,Demand/ ;
```

Define set names

Text comments, Optional in command



Assign elements to the sets

Set Definitions

Another example:

```
SET    SECTORS    sectors of the economy
   / Steel        steel mining sector (in millions of tons sold)
     Energy       energy sector (in millions of btus sold)
     Coal         coal sector (in millions of tons sold)
  / ;
```

Element explanatory text

Note: the explanatory text must not exceed 80 characters and must all be contained on the same line as the identifier it describes.

Set Definitions- Alias

ALIAS is used to give another name to previously defined sets.

```
ALIAS  (Commodity,Commodities);
```

"**Commodities**" is like a **j** and **j'** in mathematical notation.

Data Entry

Data are entered via three different types of GAMS commands

1) Scalar – for items that are not set dependent

2) Parameters – for items that are vectors (can be multidimensional)

3) Tables – for items with 2 or more dimensions

Data Entry - **SCALAR commands**

Scalar commands:

Basic format:

SCALAR ItemName optional text / value / ;

In the CGE example:

SCALAR Incometax Household tax level / 0.00 / ;

Data Entry - PARAMETER commands

Basic format:

```
PARAMETER    ItemName(setdependency)    optional text
    / element1      value1 ,
      element2      value2 /    ;
```

In the CGE example:

```
PARAMETER
    SigmaC(HouseHolds)    Household elas. of substitution
      / NonFarmer           1.5
        Farmer              0.75 /

    Phi(Sector)           Production scale parameter
      / Food                1.5
        NonFood             2.0  /          ;
```

Data Entry – TABLE commands

Basic format:

```
TABLE   ItemName(set1dep,set2dep)     optional text
                      set2elem1       set2elem2
  set1element1        value11         value12
  set1element2        value12         value22          ;
```

In our example:

```
TABLE   Intercept(Curvetype,Commodities)     intercept term
                      Corn            Wheat
Demand                6               8
Supply                1               2          ;
```

Elements from Commodities set (2nd set)

Elements from Curvetype set (1st set)

Data Entry – TABLE commands

More than two dimensional data entry using TABLE



Formulation – Variable Declarations

Basic format:

VARIALBE VarName1(setdependency) optional text

VarName2(setdependency) optional text

… ;

to declare variables < or ≥ 0

Or

POSITIVE VARIABLE

VarName1(setdependency) optional text

VarName2(setdependency) optional text

… ;

To declare ≥ 0 variables

Formulation – Variable Declarations

In our example:

```
POSITIVE VARIABLES
     P(Commodities)              Equilibrium price
     Qd(Commodities)             Quantity demanded
     Qs(Commodities)             Quantity supply    ;
```

Note that this defines a variable for each case in the set commodities and thus encompasses the cases:

P_c, P_w, Qd_c, Qd_w, Qs_c, Qs_w $\geq$ 0

Formulation – Equation Declarations

Basic format:

```
Equation      EqName1(setdependency) optional text
              EqName2(setdependency) optional text
... ;
```

Formulation – **Equation Declarations**

In our example:

```
EQUATIONS
     PDemand(Commodities)       Demand equation
     PSupply(Commodities)       Supply equation
     Equilibrium(Commodities)   Equilibrium equation ;
```

Note that this defines an equation for each case in the set commodities

Formulation – Equation Specifications

Algebraic Structure

- **Demand:** $P_c \geq Pd_c = 6 - 0.3*Qd_c - 0.1*Qd_w$

  $P_w \geq Pd_w = 8 - 0.07*Qd_c - 0.4*Qd_w$

```
Pdemand(commodities)..
     P(commodities)
      =G=
     intercepts("demand",commodities)
      + SUM(commodity,
            slopes("demand",commodities,commodity)
              *Qd(commodity));
```

Quotes " " are used to select a specific set elements.

Recall: ALIAS(commodity,commodities);

Summation Digression





Formulation – Equation Specifications

Algebraic Structure

- **Supply:** $Ps_c = 1 + 0.5*Qs_c + 0.1*Qs_w \geq P_c$

  $Ps_w = 2 + 0.1*Qs_c + 0.3*Qs_w \geq P_w$

```
Psupply(commodities)..
    intercepts("supply",commodities)
     + SUM(commodity,
         slopes("supply",commodities,commodity)
           *Qs(commodity))
    =G=
   P(commodities);
```

Formulation – Equation Specifications

Algebraic Structure

- **Equilibrium:** $Qs_c \geq Qd_c$

  $Qs_w \geq Qd_w$

```
Equilibrium(commodities)..
    Qs(commodities)
     =G=
    Qd(commodities)   ;
```

Formulation – Model and Solve Statement

```
MODEL PROBLEM   /Pdemand.Qd
                 Psupply.Qs
                 Equilibrium.P/ ;

SOLVE PROBLEM USING MCP;
```

Recall: MCP Requirements

- consistent dimension (sets) of complementary variables and equations
- no variable is complementary with more than one equation or vice versa
- every variable and equation has a complementary partner

```
POSITIVE VARIABLES
     P(Commodities)
     Qd(Commodities)
     Qs(Commodities)

EQUATIONS
     PDemand(Commodities)
     PSupply(Commodities)
     Equilibrium(Commodities)
```

Solution

```
---- EQU PDemand  Demand equation

          LOWER      LEVEL      UPPER     MARGINAL

Corn        6.000      6.000      +INF        4.373
Wheat       8.000      8.000      +INF        7.510

---- EQU PSupply  Supply equation

          LOWER      LEVEL      UPPER     MARGINAL

Corn       -1.000     -1.000      +INF        4.373
Wheat      -2.000     -2.000      +INF        7.510

---- EQU Equilibrium  Equilibrium equation

          LOWER      LEVEL      UPPER     MARGINAL

Corn          .          .       +INF        3.937
Wheat         .          .       +INF        4.690
```

Solution

```
---- VAR P  Equilibrium price

          LOWER      LEVEL      UPPER     MARGINAL

Corn        .          3.937    +INF         .
Wheat       .          4.690    +INF         .

---- VAR Qd  Quantity demanded

          LOWER      LEVEL      UPPER     MARGINAL

Corn        .          4.373    +INF         .
Wheat       .          7.510    +INF         .

---- VAR Qs  Quantity supply

          LOWER      LEVEL      UPPER     MARGINAL

Corn        .          4.373    +INF         .
Wheat       .          7.510    +INF         .
```